

Laboratorijska vježba 4

- Zadatke rješavajte u Excel-ovom VBA okruženju u okviru jednog programskog modula.

1. Napisati proceduru Zdravo koja se pokreće otvaranjem Excel radne sveske u kojoj se procedura nalazi i kojom se:
 - Od korisnika traži unos imena i prezimena.
 - Prikazuje MsgBox sa porukom kojom se pozdravlja korisnik i prikazuje mu se datum i vrijeme.
2. Napisati proceduru NovaSveska kojom se:
 - Kreira nova radna sveska.
 - Prvom radnom listu kreirane radne sveske daje ime Prvi.
 - U ćelije opsega B1:B5 lista Prvi upisuju stringovi Novembar1, Novembar2, Novembar3, Novembar4 i Novembar5 boldovanim slovima veličine 14.
 - Ćelijama iz opsega D2:F5 dodjeljuje boja čiji je indeks 21.
 - Snima kreiranu radnu svesku pod imenom Sveska u tekućem folderu.
3. Napisati proceduru koja će pri dodavanju novog radnog lista dati ime (koje korisnik unese pomoću input box-a) tom listu, pozicionira ga posle svih postojećih radnih listova. Nakon toga, u ćelije A1:A10 novog radnog lista procedura treba da upiše kvadrate prvih deset prirodnih brojeva, tj. u ćeliju A1 upisati 1, u ćeliju A2 upisati 4, ..., u ćeliju A10 upisati 100.
4. Napisati proceduru koja će se izvršiti kada korisnik pritisne desni taster miša u Sheet1 i koja prikazuje MsgBox sa imenima svih radnih listova u toj radnoj svesci.
5. Napisati proceduru koja će se izvršiti kada se promeni selekcija u listu Sheet1 i koja daje MsgBox koji prikazuje koliko nova selekcija ima ćelija, koliko ima praznih ćelija, koliko je brojeva u selekciji pozitivno, koliko negativno i koliko ima nula.
6. Napisati makro koji se aktivira desnim klikom miša na opseg unutar radnog lista. Makro treba da od korisnika preko input box-a zatraži unos stringa S i potom da prođe kroz sve ćelije selektovanog opsega i ispita da li se u riječima, upisanim u ćelijama, nalazi string S. Ako je riječ u ćeliji identična unešenom stringu, ćeliju treba obojiti u crveno i tekst boldovati, a ako string u ćeliji sadrži string S ćeliju obojiti u zeleno.

Rješenja

1)

U okviru modula:

```
Public Sub zdravo()
```

```
    Dim imePrezime As String
```

```
    Dim poruka As String
```

```
    imePrezime = InputBox("Molimo unesite vaše ime i prezime", "Pozdrav")
```

```
    poruka = "Zdravo, " & imePrezime & "! " & vbCrLf & "Trenutni datum i vrijeme je: " & Date & Time
```

```
    MsgBox poruka, vbInformation, "Dobrodošlica"
```

```
End Sub
```

U okviru ThisWorkbook-a:

```
Private Sub Workbook_Open()
```

```
    zdravo
```

```
End Sub
```

2)

```
Sub NovaSveska()
```

```
    Dim sveska As Workbook
```

```
    Dim list As Worksheet
```

```
    Dim celija As Range
```

```
    Dim i As Integer
```

```
    Set sveska = Workbooks.Add
```

```
    Set list = sveska.Sheets(1)
```

```
    list.Name = "Prvi"
```

```
    i = 1
```

```
    For Each celija In list.Range("B1:B5")
```

```
        With celija
```

```
            .Value = "Novembar" & i
```

```
            With .Font
```

```
                .Bold = True
```

```
                .Size = 14
```

```
            End With
```

```
        End With
```

```
        i = i + 1
```

```
    Next celija
```

```
    With list.Range("D2:F5").Interior
```

```
        .ColorIndex = 21
```

```
    End With
```

```
    sveska.SaveAs ThisWorkbook.Path & "\Sveska.xlsx"
```

```
End Sub
```

3)

```
Private Sub Workbook_NewSheet(ByVal Sh As Object)
```

```
Dim ime As String, L As Worksheet, brojListova As Integer

ime = InputBox("Uneti ime: ", "Ime radnog lista")
brojListova = ThisWorkbook.Worksheets.Count
Sh.Name = ime
Sh.Move After:=ThisWorkbook.Worksheets(brojListova)
For i = 1 To 10
    Sh.Cells(i, 1).Value = i ^ 2
Next i
End Sub

4)
Private Sub Worksheet_BeforeRightClick(ByVal Target As Range, Cancel As Boolean)
    Dim list As Worksheet
    Dim listalmena As String

    For Each list In ThisWorkbook.Sheets
        listalmena = listalmena & list.Name & vbCrLf
    Next list

    MsgBox "Imena radnih listova radne sveske su:" & vbCrLf & listalmena, vbInformation, "Lista Radnih Listova"

    Cancel = True
End Sub

5)
Private Sub Worksheet_SelectionChange(ByVal Target As Range)
    Dim Poz As Integer, Neg As Integer, Nule As Integer
    Dim Prazne As Integer, S As String, Celija As Range
    Prazne = 0: Poz = 0: Neg = 0: Nule = 0

    For Each Celija In Target.Cells
        If Celija.Value = "" Then
            Prazne = Prazne + 1
        ElseIf IsNumeric(Celija.Value) Then
            If Celija.Value > 0 Then
                Poz = Poz + 1
            ElseIf Celija.Value < 0 Then
                Neg = Neg + 1
            Else
                Nule = Nule + 1
            End If
        End If
    Next

    S = "Ima " & Target.Count & " celija, od toga:" & vbCrLf
    S = S & Prazne & " praznih celija," & vbCrLf
    S = S & Poz & " pozitivnih brojeva," & vbCrLf
    S = S & Neg & " negativnih brojeva i " & vbCrLf & Nule & " nula."
    MsgBox S

End Sub

6)
Private Sub Worksheet_BeforeRightClick(ByVal Target As Range, Cancel As Boolean)
    Dim trazeniString As String
    Dim celija As Range
```

```
trazeniString = InputBox("Unesite string za pretragu:", "Pretraga Stringa")

If trazeniString = "" Then Exit Sub

For Each celija In Selection
    If IsNumeric(celija.Value) Or IsEmpty(celija.Value) Then
        GoTo NextCell
    End If

    If celija.Value = trazeniString Then
        With celija.Interior
            .Color = RGB(255, 0, 0) ' Bojimo celiju u crveno
        End With
        celija.Font.Bold = True ' Tekst postaje bold

    ElseIf InStr(1, celija.Value, trazeniString, vbTextCompare) > 0 Then
        celija.Interior.Color = RGB(0, 255, 0) ' Bojimo celiju u zeleno
    End If

NextCell:
Next celija
Cancel = True
End Sub
```